

INTRODUCTION TO VERILOG

Objectives:

In this lab, student will be able to

1. Learn the basic concepts of logic circuits and analyze the logic network.
2. Write the Truth table and Timing diagram.
3. Understand different representation of logic circuits in Verilog.
4. Learn the different tools available in the CAD system.
5. Write and simulate logic circuits using Verilog.

I. Basic concepts of Logic Circuits

Logic Circuits

- Perform operations on digital signals.
- Signal values are restricted to a few discrete values.
- In binary logic circuits, there are only two values, 0 and 1.

Logic Gates and Networks

- Each logic operation can be implemented electronically with transistors, resulting in a circuit element called a logic gate.
- It has one or more inputs and one output that is a function of its inputs.
- It is often convenient to describe a logic circuit by drawing a circuit diagram, or schematic, consisting of graphical symbols representing the logic gates.

The graphical symbols for the AND, NOT and OR gates are shown in Fig. 1.1, 1.2 and 1.3 respectively.




Figure 1.1




Figure 1.2




Figure 1.3

- A larger circuit is implemented by a network of gates, as shown in Fig. 1.4




Figure 1.4

x_1	x_2	$x_1 \cdot x_2$	$x_1 + x_2$
0	0	0	0
0	1	0	1
1	0	0	1
1	1	1	1
		AND	OR

Figure 1.5 Truth Table

- The operations AND, OR etc can also be defined in the form of a table as shown in Figure 1.5.
- The first two columns (to the left of the heavy vertical line) give all four possible combinations of logic values that the variables x_1 and x_2 can have.
- The next column defines the AND operation for each combination of values of x_1 and x_2 , and the last column defines the OR operation.
- In general, for n input variables the truth table has 2^n rows.

II. Analysis of a Logic Network

- Determining the function performed by an existing logic network is referred to as the Analysis process.
- The reverse task of designing a new network that implements a desired functional behavior is referred to as the Synthesis process.
- To determine the functional behavior of the network in Fig. 1.6, we can consider what happens if we apply all possible values to input signals x_1 and x_2 . The analysis of these input values at various intermediate points is shown in Fig. 1.7.




Figure 1.6

x1	x2	A	B	f
0	0	1	0	1
0	1	1	0	1
1	0	0	0	0
1	1	0	1	1

Figure 1.7

Timing Diagram

- The information in Fig. 1.7 can be presented in graphical form, known as a timing diagram, as shown in Fig. 1.8.
- The figure shows the waveforms for the inputs and output of the network, as well as for the internal signals at the points labeled A and B.




Figure 1.8

Functionally Equivalent Networks

- Going through the same analysis procedure, we find that the output 'g' in Fig. 1.9, changes in exactly the same way as f does in Fig. 1.6.
- Therefore, $g(x_1, x_2) = f(x_1, x_2)$, which indicates that the two networks are functionally equivalent.




Figure 1.9

III. Introduction to CAD Tools

- Logic circuits are designed using CAD tools that automatically implement synthesis techniques.
- CAD system includes tools for design entry, synthesis and optimization, simulation and physical design.

Design Entry

- The starting point in the process of designing a logic circuit is the conception of what the circuit is supposed to do and the formulation of its general structure.

- The first stage of this process involves entering into the CAD system a description of the circuit being designed. This stage is called design entry.
- For design entry, we are writing source code in a hardware description language.

Hardware Description Languages

- A hardware description language (HDL) is similar to a typical computer programming language except that an HDL is used to describe hardware rather than a program to be executed on a computer.
- Two HDLs are IEEE standards: Verilog HDL and VHDL.

Why use Verilog

- Supported by most companies that offer digital hardware technology.
- Verilog provides design portability. A circuit specified in Verilog can be implemented in different types of chips and with CAD tools provided by different companies, without changing the Verilog specification.
- Both small and large logic circuit designs can be efficiently represented in Verilog code.

Functional Simulation

- The functional simulator tool verifies that the designed circuit functions as expected.
- It uses two types of information.
 - First, the user's initial design is represented by the logic equations generated during synthesis.
 - Second, the user specifies the valuations of the circuit's inputs that should be applied to these equations during the simulation.
- For each valuation, the simulator evaluates the outputs produced by the expressions.
- The results of simulations are usually provided in the form of a timing diagram that the user can examine to verify that the circuit operates as required.

Timing Simulation

- When the values of inputs to the circuit change it takes a certain amount of time before a corresponding change occurs at the output. This is called a propagation delay of the circuit.

IV. Representation of Digital Circuits in Verilog

- **Structural representation-** A larger circuit is defined by writing code that connects simple circuit elements together.
- **Behavioral representation-** Describing a circuit by using logical expressions and programming constructs that define the behavior of the circuit but not its actual structure in terms of gates.

Structural Specification of Logic Circuits

- A gate is represented by indicating its functional name, output, and inputs. Different logic gates are shown in Table 1.1

For example,

- A two-input AND gate, with inputs x1 and x2 and output y, is denoted as
and (y, x1, x2);
- A four-input OR gate is specified as
or (y, x1, x2, x3, x4);
- The NOT gate is given by **not** (y, x); implements $y = x'$.

Table 1.1

Name	Description	Usage
and	$f = (a \cdot b \cdot \dots)$	and (f, a, b, ...)
nand	$f = \overline{(a \cdot b \cdot \dots)}$	nand (f, a, b, ...)
or	$f = (a + b + \dots)$	or (f, a, b, ...)
nor	$f = \overline{(a + b + \dots)}$	nor (f, a, b, ...)
xor	$f = (a \oplus b \oplus \dots)$	xor (f, a, b, ...)
xnor	$f = (a \odot b \odot \dots)$	xnor (f, a, b, ...)
not	$f = \overline{a}$	not (f, a)

Verilog Module

- It is a circuit or subcircuit described with Verilog code.
- The module has a name, **module_name**, which can be any valid identifier, followed by a list of ports.
- The term port refers to an input or output connection in an electrical circuit. The ports can be of type **input**, **output**, or **inout** (bidirectional), and can be either scalar or vector.

The General Form of a Module

```
module module name [(port name{, port name})];  
    [parameter declarations]  
    [input declarations]  
    [output declarations]  
    [inout declarations]  
    [wire or tri declarations]  
    [reg or integer declarations]
```

```

[function or task declarations]
[assign continuous assignments]
[initial block]
[always blocks]
[gate instantiations]
[module instantiations]
endmodule

```

Documentation in Verilog Code

- Documentation can be included in Verilog code by writing a comment. A short comment begins with the double slash, //, and continues to the end of the line. A long comment can span multiple lines and is contained inside the delimiters /* and */.

White Space

- White space characters, such as SPACE and TAB, and blank lines are ignored by the Verilog compiler.
- Multiple statements can be written on a single line.
- Placing each statement on a separate line and using indentation within blocks of code, such as an **if-else** statement are good ways to increase the readability of code.

Signals in Verilog Code

- A signal in a circuit is represented as a net or a variable with a specific type.
- A net or variable declaration has the form
type [range] signal_name{, signal_name};
- The signal_name is an identifier
- The range is used to specify vectors that correspond to multi-bit signals

Signal Values and Numbers

- Verilog supports scalar nets and variables that represent individual signals and vectors that correspond to multiple signals.
- Each individual signal can have four possible values:

0 = logic value 0	1 = logic value 1
z = tri-state (high impedance)	x = unknown value
- The value of a vector variable is specified by giving a constant of the form [size][radix]constant where size is the number of bits in the constant, and radix is the number base. Supported radices are

d = decimal	b = binary	h = hexadecimal	o = octal
-------------	------------	-----------------	-----------
- Some examples of constants include

0 the number 0	10 the decimal number 10
'b10 the binary number 10 = (2) ₁₀	'h10 the hex number 10 = (16) ₁₀
4'b100 the binary number 0100 = (4) ₁₀	

Nets

Verilog defines a number of types of nets.

- A net represents a node in a circuit.
- For synthesis purposes, the only important nets are of **wire** type.
- For specifying signals that are neither inputs nor outputs of a module, which are used only for internal connections within the module, Verilog provides the **wire** type.

Identifier Names

- Identifiers are the names of variables and other elements in Verilog code.
- The rules for specifying identifiers are simple: any letter or digit may be used, as well as the _ underscore and \$ characters.
- An identifier must not begin with a digit and it should not be a Verilog keyword.
 - Examples of legal identifiers are f, x1, x, y, and Byte.
 - Some examples of illegal names are 1x, +y, x*y, and 258
- Verilog is case sensitive, hence k is not the same as K, and BYTE is not the same as Byte.

Verilog Operators

- Verilog operators are useful for synthesizing logic circuits.
- Table 1.2 lists these operators in groups that reflect the type of operation performed.

Table 1.2

Operator type	Operator Symbols	Operation Performed	Number of operands
Bitwise	~	1's complement	1
	&	Bitwise AND	2
		Bitwise OR	2
	^	Bitwise XOR	2
	~^ or ^~	Bitwise XNOR	2
Logical	!	NOT	1
	&&	AND	2
		OR	2
Reduction	&	Reduction AND	1
	~&	Reduction NAND	1
		Reduction OR	1
	~	Reduction NOR	1
	^	Reduction XOR	1
	~^ or ^~	Reduction XNOR	1
Arithmetic	+	Addition	2

	-	Subtraction	2
	-	2's complement	1
	*	Multiplication	2
	/	Division	2
Relational	>	Greater than	2
	<	Lesser than	2
	>=	Greater than or equal to	2
	<=	Lesser than or equal to	2
Equality	==	Logical equality	2
	!=	Logical inequality	2
Shift	>>	Right shift	2
	<<	Left shift	2
Concatenation	{,}	Concatenation	Any number
Replication	{{}}	Replication	Any number
Conditional	?:	Conditional	3

Running a sample Verilog code

Let us look at a Verilog implementation in the following steps:

1. Create a directory with section followed by roll number(to be unique); e.g. A21
2. **Start → Programs → Altera → MAX + plus II**
3. The MAX + plus II manager window will pop up.
4. Select **File** from the menu bar, then select **Project → Name** and give project name. The project name will come on the title bar. This step is optional. If skipped, before compilation follow **File → Project → Set Project to Current File**.
5. Then follow **MAX + plus II → Text Editor**. Text Editor is used to typing the source code.

```


module example2(x1,x2,x3,f);
    input x1,x2,x3;
    output f;
    and (g,x1,x2);
    not (k,x2);
    and (h,k,x3);
    or (f,g,h);
endmodule

```

6. Save the source code with file name same as project name but with **‘.v’** extension in the required directory.

7. Then follow **MAX + plus II → Compiler**. If step 2 is skipped, follow **File → Project → Set Project to Current File** first and then select Compiler. The compiler window will pop up.
8. Click **Start**. The source file will be compiled and errors if any will be displayed. Double click on the first error to rectify it. Rectify all the errors and recompile to get an error-free compiled file.
9. Select **MAX + plus II → Waveform Editor**. The Waveform Editor window will pop up. Then select **Node → Enter Nodes from SNF (Simulator Netlist File)** and then in the popped-up box click **List** button to display the names of input and output nodes in the box labeled **Available Nodes & Groups**. Select the nodes if not selected, and click => button to copy them into **Selected Nodes & Groups** box. Click **OK** to return to the waveform editor. The nodes are now displayed in the waveform display.
10. Now the values to the inputs will be given. Select **File → End Time** to specify the total amount of time for simulation. If not specified it will take default time.
11. Select **View → Fit in Window** so that the entire time range is visible in the waveform editor display. Select **Options → Grid Size** to give an appropriate grid size.
12. Either the single value or different values at different grids can be given to the input signals. To give a single value select the entire signal and click the required value which is activated on the left side of the window. To give different values select a section of the signal by dragging the mouse over it and then by clicking the required value. After giving the values save the file with the same name as before but with .scf (Simulator Channel File) extension.
13. Select **MAX + plus II → Simulator**. Click **Start**. A message box will be displayed indicating no errors. Click **OK**. And then **Open SCF** to view the waveforms after simulation.

Output:



V. Behavioral Specification of Logic Circuits

- Gate level primitives can be tedious when large circuits have to be designed.

- Abstract expressions and programming constructs are used to describe the behavior of a digital circuit.
- To define the circuit using logic expressions. The AND and OR operations are indicated by the ‘&’ and ‘|’ signs, respectively.
- The **assign** keyword provides a continuous assignment for the output signal.
- Whenever any signal on the right-hand side changes its state, the value of output will be re-evaluated.

```

module example2 (x1, x2, x3, f);
    input x1, x2, x3;
    output f;
    assign f = (x1 & x2) | (x2 & x3);
endmodule

```

SIMPLIFICATION USING K-MAP

Objectives:

In this lab, student will be able to

1. Understand the steps for optimization using K-map.
2. Design minimum cost circuit.
3. Write Verilog code for the simplified expression.

I. Steps for Optimization using K-map:

K-map

- A systematic way of performing optimization.
- Finds a minimum-cost expression for a given logic function by reducing the number of product (or sum) terms needed in the expression, by applying the combining property.
- **Implicant-** A product term that indicates the input valuation(s) for which a given function is equal to 1.
- **Prime Implicant-** An implicant is called a prime implicant if it cannot be combined into another implicant that has fewer literals.
- **Essential prime implicant-** If a prime implicant includes a minterm for which $f=1$ that is not included in any other prime implicant, then it must be included in the cover and is called as an essential prime implicant.
- The process of finding a minimum-cost circuit involves the following steps:
 1. Generate all prime implicants for the given function f .
 2. Find the set of essential prime implicants.

3. If the set of essential prime implicants covers all valuations for which $f = 1$, then this set is the desired cover of f . Otherwise, determine the nonessential prime implicants that should be added to form a complete minimum-cost cover.

II. Incompletely Specified Functions

- A function that has don't-care condition(s).
- Using the shorthand notation, the function f is specified as

$$f(x_1, \dots, x_4) = \sum m(2, 4, 5, 6, 10) + D(12, 13, 14, 15)$$
 where D is the set of don't cares.

Solved Exercise:

Simplify the following function using K-map and write Verilog code to implement this.

$$f(x_1, \dots, x_4) = \sum m(2, 4, 5, 6, 10) + D(12, 13, 14, 15)$$

x_1x_2		x_3x_4			
		00	01	11	10
00	0	1	d	0	$x_2\bar{x}_3$
01	0	1	d	0	
11	0	0	d	0	$x_3\bar{x}_4$
10	1	1	d	1	

$$f = x_2\bar{x}_3 + x_3\bar{x}_4$$

Verilog code:

```

module example4(x2,x3,x4,f);
  input x2,x3,x4;
  output f;
  assign f=(x2 & ~x3) | (x3 & ~x4);
endmodule

```

MULTILEVEL SYNTHESIS

Objectives:

In this lab, student will be able to

1. Design multilevel NAND and NOR circuits.
2. Write Verilog code for multilevel circuits.
3. Use functional decomposition for the synthesis of multilevel circuits.

- **Multilevel NAND and NOR Circuits**

- Multilevel AND-OR circuits can be realized by a circuit that contains only NAND gates or only NOR gates.
- Each AND gate is converted to a NAND by inverting its output.
- Each OR gate is converted to a NAND by inverting its inputs.
- Each AND gate is converted to a NOR by inverting its inputs.
- Each OR gate is converted to a NOR by inverting its output.
- Inversions that are not a part of any gate can be implemented as two-input NAND/NOR gates, where the inputs are tied together.
- **Functional decomposition-** Complex logic circuit can be reduced by decomposing a two-level circuit into subcircuits, where one or more subcircuits implement functions that may be used in several places to construct the final circuit.

Solved Exercise

Apply functional decomposition for the following function to obtain a simplified circuit and simulate using Verilog.

$$f = \bar{x}_1 x_2 x_3 + x_1 \bar{x}_2 x_3 + x_1 x_2 x_4 + \bar{x}_1 \bar{x}_2 x_4$$

Factoring x_3 from the first two terms and x_4 from the last two terms, this expression becomes


$$f = (\bar{x}_1 x_2 + x_1 \bar{x}_2) x_3 + (x_1 x_2 + \bar{x}_1 \bar{x}_2) x_4$$

Now let $g(x_1, x_2) = \bar{x}_1 x_2 + x_1 \bar{x}_2$ and observe that

$$\begin{aligned} \bar{g} &= \overline{\bar{x}_1 x_2 + x_1 \bar{x}_2} \\ &= \overline{\bar{x}_1 x_2} \cdot \overline{x_1 \bar{x}_2} \\ &= (x_1 + \bar{x}_2)(\bar{x}_1 + x_2) \\ &= x_1 \bar{x}_1 + x_1 x_2 + \bar{x}_2 \bar{x}_1 + \bar{x}_2 x_2 \\ &= 0 + x_1 x_2 + \bar{x}_1 \bar{x}_2 + 0 \\ &= x_1 x_2 + \bar{x}_1 \bar{x}_2 \end{aligned}$$

Then f can be written as

$$f = g x_3 + \bar{g} x_4$$



Verilog code:

```

module example5(x1,x2,x3,x4,f);
    input x1,x2,x3,x4;
    output f;
    assign g=(x1 & ~x2) | (~x1 & x2);
    assign f=(g & x3) | (~g & x4);
endmodule

```

ARITHMETIC CIRCUITS

Objectives:

In this lab, student will be able to

1. Design arithmetic circuits using combinational logic.
2. Simulate arithmetic circuits using Verilog.

I. Adder circuit:

- **Half adder**- a circuit that implements the addition of only two single bit inputs.
- **Full adder**- a circuit that implements the addition of two single bit inputs and one carry bit.
- **Ripple-carry adder**
 - For each bit position, we can use a full-adder circuit, connected as shown in Fig. 4.1.
 - Carries that are produced by the full-adders propagate to the left.




Figure 4.1 An n-bit ripple carry adder

- **Adder/Subtractor unit**-
 - The only difference between performing addition and subtraction is that for subtraction it is necessary to use the 2's complement of one operand.
 - Add/Sub control signal chooses whether addition or subtraction is to be performed.
 - Outputs of the XOR gates represent Y if Add/Sub = 0, and they represent the 1's complement of Y if Add/Sub = 1.

- Add/Sub is also connected to the carry-in c_0 . This makes $c_0 = 1$ when subtraction is to be performed, thus adding the 1 that is needed to form the 2's complement of Y .
- When the addition operation is performed, we will have $c_0 = 0$.
- The circuit is shown in Fig. 4.2




Figure 4.2 Adder/subtractor unit

- **Binary multiplier**

- Multiplication of binary numbers is performed in the same way as in decimal numbers. The multiplicand is multiplied by each bit of the multiplier starting from the least significant bit.
- Each such multiplication forms a partial product. Successive partial products are shifted one position to the left. The final product is obtained from the sum of the partial products.

- **BCD Addition**

In Binary Coded Decimal (BCD) representation each digit of a decimal number is represented by 4-bit binary. When 2 BCD numbers are added,

- If $X + Y \leq 9$, then the addition is the same as the addition of 2 four-bit unsigned binary numbers.
- A correct decimal digit can be generated by adding 6 to the result of a four-bit addition whenever the result exceeds 9 or when the carry is generated.

II. Designing subcircuits in Verilog

- A Verilog module can be included as a subcircuit in another module.
- Both modules must be defined in the same file.
- The general form of a module instantiation statement is given below.

```
module_name [#(parameter overrides)] instance_name (
    .port_name ( [expression] ) { , .port_name ( [expression] ) } );
```

- The ***instance_name*** can be any legal Verilog identifier and the port connections specify how the module is connected to the rest of the circuit.
- The same module can be instantiated multiple times in a given design provided that each instance name is unique.
- The #(parameter overrides) can be used to set the values of parameters defined inside the ***module_name*** module.
- Each ***port_name*** is the name of a port in the subcircuit, and each expression specifies a connection to that port.
- **Named port connections** -The syntax ***.port_name*** is provided so that the order of signals listed in the instantiation statement does not have to be the same as the order of the ports given in the **module** statement of the subcircuit.
- **Ordered port connections**-If the port connections are given in the same order as in the subcircuit, then ***.port_name*** is not needed.

Using Vectored Signals

- Multibit signals are called *vectors*.
- An example of an input vector is
input [3:0] W;
- This statement defines *W* to be a four-bit vector. Its individual bits can be referred to using an index value in square brackets.
- The most-significant bit (MSB) is referred to as *W[3]* and the least-significant bit (LSB) is *W[0]*.

Solved Exercise

Write the Verilog code to implement a 4-bit adder.

Verilog code:

```

module adder4 (carryin, x3, x2, x1, x0, y3, y2, y1, y0, s3, s2, s1, s0, carryout);
    input carryin, x3, x2, x1, x0, y3, y2, y1, y0;
    output s3, s2, s1, s0, carryout;
    fulladd stage0 (carryin, x0, y0, s0, c1);
    fulladd stage1 (c1, x1, y1, s1, c2);
    fulladd stage2 (c2, x2, y2, s2, c3);
    fulladd stage3 (c3, x3, y3, s3, carryout);
endmodule

module fulladd (Cin, x, y, s, Cout);
    input Cin, x, y;
    output s, Cout;
    assign s = x ^ y ^ Cin;

```

```
assign Cout = (x & y) | (x & Cin) | (y & Cin);  
endmodule
```

COMPARATORS AND CODE CONVERTERS

Objectives:

In this lab, student will be able to

1. Learn the concept of comparators and code converters.
2. Write Verilog code to simulate comparators and code converters.

Arithmetic Comparison Circuits

- Compare the relative magnitude of two binary numbers.

Code Converters

- Convert from one type of input representation to a different output representation.

Parameters

- A parameter associates an identifier name with a constant.
- Using the following declaration, the identifier `n` can be used in place of the number 4.
parameter n = 4;

Verilog Procedural Statements

- Rather than using gates or logic expressions, circuits can be specified in terms of their behavior.
- Also called sequential statements.
- Procedural statements are evaluated in the order in which they appear in the code whereas concurrent statements are executed in parallel.
- Verilog syntax requires that procedural statements should be inside an **always** block.

Always Block

- An **always** block is a construct that contains one or more procedural statements.
- It has the form


```
always @(sensitivity_list)
[begin]
    [procedural assignment statements]
    [if-else statements]
    [case statements]
    [while, repeat, and for loops]
    [task and function calls]
[end]
```

- The *sensitivity_list* is a list of signals that directly affect the output results generated by the **always** block.
- If the value of a signal in the sensitivity list changes, then the statements inside the **always** block are evaluated in the order presented.

Variables:

- A variable can be assigned a value and this value is retained until it is overwritten in a subsequent assignment statement.
- There are two types of variables, **reg**, and an **integer**.
 - The keyword **reg** does not denote a storage element or register. In Verilog code, **reg** variables can be used to model either combinational or sequential parts of a circuit.
 - **Integer** variables are useful for describing the behavior of a module, but they do not directly correspond to nodes in a circuit.

The if-else Statement

- The general form of the **if-else** statement is given below.

```
if (expression1)
begin
    statement;
end
else if (expression2)
begin
    statement;
end
else
begin
    statement;
end
```

- If expression1 is true, then the first statement is evaluated.

- When multiple statements are involved, they have to be included inside a **begin-end** block.
- The **else if** and **else** clauses are optional.
- Verilog syntax specifies that when **else if** or **else** are included, they are paired with the most recent unfinished **if** or **else if**.

for Loop

- the general form of **for** loop

```
for (initial_index; terminal_index; increment)
begin
    statement;
end
```

- The ***initial_index*** is evaluated once, before the first loop iteration, and typically performs the initialization of the **integer** loop control variable.
- In each loop iteration, the begin-end block is performed, and then the increment statement is evaluated.
- Finally, the ***terminal_index*** condition is checked, and if it is True (1), then another loop iteration is done.

Solved Exercise

Write Verilog code to simulate a 1-bit equality comparator.

Verilog code:

```
module compare(A, B, AeqB);
    input A, B;
    output reg AeqB;
    always @(A,B)
    begin
        AeqB=0;
        if (A==B)
            AeqB=1;
    end
endmodule
```

MULTIPLEXERS

Objectives:

In this lab, student will be able to

1. Understand the concept of multiplexers.
2. Learn more about the behavioral style of Verilog programming.
3. Design and implement simple multiplexers.
4. Design and implement large multiplexers using small multiplexers.

Multiplexers

- The multiplexer has a number of data inputs, one or more select inputs, and one output.
- It passes the signal value on one of the data inputs to the output.
- A multiplexer that has N data inputs, w_0, \dots, w_{N-1} , requires $\log_2 N$ select inputs.
- Fig. 6.1a shows the graphical symbol for a 2-to-1 multiplexer.
- The functionality of a multiplexer can be described in the form of a truth table. Fig. 6.1b shows the functionality of a 2-to-1 multiplexer.




Figure 6.1a Graphical symbol

s	f
0	w_0
1	w_1

Figure 6.1b Truth table

The Conditional Operator

- In a logic circuit, it is often necessary to choose between several possible signals or values based on the state of some condition.
- Verilog provides a conditional operator ($?:$) which assigns one of the two values depending on a conditional expression.

Syntax of conditional operator

```
conditional_expression ? true_expression : false_expression
```

- If the conditional expression evaluates to 1 (true), then the value of true_expression is chosen; otherwise, the value of false_expression is chosen.

The case Statement

- The general form of a **case** statement is given below.

```

case (expression)
    alternative1: begin
        statement;
    end
    alternative2: begin
        statement;
    end
    [default: begin
        statement;
    end]
endcase

```

- The bits in expression, called as controlling expression, are checked for a match with each alternative.
- The first successful match causes the associated statements to be evaluated.
- Each digit in each alternative is compared for an exact match of the four values 0, 1, x, and z.
- A special case is the **default** clause, which takes effect if no other alternative matches.
- The **casex** statement reads all z and x values as don't cares.

Solved Exercise

Write behavioral Verilog code for 2 to 1 multiplexer using **always** and **conditional** operators.

Verilog code:

```

module mux2to1 (w0, w1, s, f);
    input w0, w1, s;
    output f;
    reg f;
    always @(w0 or w1 or s)
        f = s ? w1 : w0;
endmodule

```

MULTIPLEXER APPLICATIONS

Objectives:

In this lab, student will be able to

1. Learn how to synthesis logic functions using multiplexers.
2. Write Verilog code to synthesis logic functions using multiplexers.

Synthesis of Logic Functions Using Multiplexers

- Procedure to synthesis a logic function is as shown in Fig. 7.1

- One of the input signals, w_1 in this function, is chosen as the select input to the 2 to 1 multiplexer
- When $w_1 = 0$, f has the same value as input w_2 , and when $w_1 = 1$, f has the value of w_2' .

Solved Exercise

Realize the function $f = w_1 \oplus w_2$ using a 2 to 1 multiplexer and other necessary gates. Write a Verilog code to implement the design.

w_1	w_2	f		w_1	f
0	0	0	}	0	w_2
0	1	1		0	\bar{w}_2
1	0	1	}	1	
1	1	0		1	

Figure 7.1a Truth table




Figure 7.1b Circuit

Verilog code:

```

module mux2to1 (w1, w2, f);
    input w1, w2;
    output f;
    reg f;
    always @(w1 or w2)
        f = w1?~w2:w2;
endmodule

```

DECODERS AND ENCODERS

Objectives:

In this lab, student will be able to

1. Learn the concept of decoders, encoders and priority encoders.
2. Write Verilog code for decoders, decoder trees and encoders.

Decoders

- Decoder circuits are used to decode the encoded information.
- A binary decoder, depicted in Fig. 8.1, is a logic circuit with n inputs and 2^n outputs.
- Each output corresponds to one valuation of the inputs, and only one output is asserted at a time.
- The decoder also has an enable input En , that is used to disable the outputs; if $En = 0$, then none of the decoder outputs is asserted.
- If $En = 1$, the valuation of $w_{n-1} \cdots w_1 w_0$ determines which of the outputs is asserted.
- Larger decoders can be built using smaller decoders referred to as decoder tree.




Figure 8.1 Decoder

Binary Encoders

- A binary encoder encodes information from 2^n inputs into an n -bit code, as indicated in Fig. 8.2.
- Exactly one of the input signals should have a value of 1, and the outputs present the binary number that identifies which input is equal to 1.




Figure 8.2 Encoder

Priority Encoder

- In a priority encoder, each input has a priority level associated with it.

- When an input with a high priority is asserted, the other inputs with lower priority are ignored. Since it is possible that none of the inputs is equal to 1, an output, z , is provided to indicate this condition.
- The truth table for a 4-to-2 priority encoder is shown in Fig. 8.3.

w_3	w_2	w_1	w_0	y_1	y_0	z
0	0	0	0	d	d	0
0	0	0	1	0	0	1
0	0	1	x	0	1	1
0	1	x	x	1	0	1
1	x	x	x	1	1	1

Figure 8.3 Truth table for a 4 to 2 priority encoder

Casex Statement:

- Verilog provides variants of the **case** statement that treat the z and x values in a different way.
- The **casez** statement treats all z values as don't cares.
- The **casex** statement treats all z and x values as don't cares.

Solved exercise

Write behavioral Verilog code for 2 to 4 binary decoder using **for** loop.

Verilog code:

```

module dec2to4 (W, Y, En);
    input [1:0] W;
    input En;
    output [0:3] Y;
    reg [0:3] Y;
    integer k;
    always @(W or En)
    for (k = 0; k <= 3; k = k+1)
    if ((W == k) && (En == 1))
        Y[k] = 1;
    else
        Y[k] = 0;
endmodule

```

APPLICATIONS OF DECODERS

Objectives:

In this lab, student will be able to

1. Implement logic functions using decoders and other necessary gates.
2. Write the Verilog code to implement logical functions using decoders.

Applications of Decoders

The decoder generates a separate output for each minterm of the required function. These outputs are combined using the OR gate as shown in Fig. 9.1.

Solved Exercise

Implement the function $f(w_1, w_2, w_3) = \sum m(0, 1, 3, 4, 6, 7)$ by using 3 to 8 binary decoder and an OR gate. Write the Verilog code to implement the same.

Solution:




Figure 9.1

Verilog code:

```
module logicfn(W, En,f);  
    input [2:0] W;  
    input En;  
    output f;  
    wire [0:7] Y;  
    dec3to8 decoder(W,Y,En);  
    assign f=Y[0] | Y[1] | Y[3] | Y[4] | Y[6] | Y[7];  
endmodule
```



```

module dec3to8 (W, Y, En);
    input [2:0] W;
    input En;
    output [0:7] Y;
    reg [0:7] Y;
    always @(W or En)
    begin
        if (En == 0)
            Y = 8'b00000000;
        else
            case (W)
                0: Y = 8'b10000000;
                1: Y = 8'b01000000;
                2: Y = 8'b00100000;
                3: Y = 8'b00010000;
                4: Y = 8'b00001000;
                5: Y = 8'b00000100;
                6: Y = 8'b00000010;
                7: Y = 8'b00000001;
            endcase
        end
    endmodule

```

FLIP FLOPS AND REGISTERS

Objectives:

In this lab. student will be able to

1. Learn different types of Flip Flops.
2. Understand the concept of triggering of flip flops and different types of reset.
3. Understand the concept of registers and shift registers.
4. Write Verilog code for Flip Flops and registers.

I. Flip Flops and Registers

Flip Flops:

- A flip flop circuit can maintain a binary state until directed by an input signal to switch the state.
- Major differences among various types of flip flops are in the number of inputs they process and in the manner in which the inputs affect the binary state.

Triggering of Flip-Flops:

- The state of a flip flop is switched by a momentary change in the input signal which is called triggering the flip flop.

Positive Edge and Negative Edge:

- A positive clock source remains at 0 during the interval between pulses and goes to 1 during the occurrence of a pulse.
- The pulse transition from 0 to 1 is called a **positive edge** and return from 1 to 0 is called a **negative edge**.

Registers:

- A register is a group of binary cells suitable for holding binary information.

Shift Registers:

- A register capable of shifting its binary information either to the right or to the left is called a shift register.

II. Verilog Constructs for Storage Elements

- The Verilog keywords **posedge** and **negedge** are used to implement edge-triggered circuits.
- The keyword **posedge** specifies that a change may occur only on the positive edge of Clock.
- The keyword **negedge** specifies that a change may occur only on the negative edge of Clock.

Blocking and Non-Blocking Assignments

Blocking

- A Verilog compiler evaluates the statements in an **always** block in the order in which they are written.
- If a variable is given a value by a blocking assignment statement, then this new value is used in evaluating all subsequent statements in the block.
- Denoted by the '=' symbol
- Example
 Q1 = D;
 Q2 = Q1;
- The above statement results in Q2=Q1=D

Non-Blocking

- Verilog also provides a non-blocking assignment, denoted with '<='.

- All non-blocking assignment statements in an **always** block are evaluated using the values that the variables have when the **always** block is entered.
- Thus, a given variable has the same value for all statements in the block.
- The meaning of non-blocking is that the result of each assignment is not seen until the end of the **always** block.
- Example


```
Q1 <= D;
Q2 <= Q1;
```
- The variables Q1 and Q2 have some value at the start of evaluating the **always** block, and then they change to a new value concurrently at the end of the **always** block.

Flip-Flops with Clear Capability

- By using a particular sensitivity list and a specific style of an **if-else** statement, it is possible to include clear (or preset) signals on flip-flops.

Solved Exercise:

Write behavioral Verilog code for positive edge-triggered D FF with synchronous reset.

Verilog Code:

```
module flipflop (D, Clock, Resetn, Q);
    input D, Clock, Resetn;
    output Q;
    reg Q;
    always @(posedge Clock)
        if (!Resetn)
            Q <= 0;
        else
            Q <= D;
endmodule
```

COUNTERS

Objectives:

In this lab, student will be able to

1. Learn the concept of synchronous/asynchronous up/down counters.
2. Learn the concept of ring and Johnson counters.
3. Write Verilog code for different types of counters.

Counters:

- A counter is essentially a register that goes through a predetermined sequence of states upon the application of input pulses.
- A binary counter with a reverse count is called a binary down counter.

Ripple Counters

- Also called as asynchronous counters.
- The CP inputs of all flip flops (except the first) are triggered not by the incoming pulses, but by the transition that occurs in other flip flops.
- Four-bit binary ripple counter is shown in Fig. 11.1

Synchronous Counters

- The input pulses are applied to the CP input of all the flip flops.
- The common pulse triggers all flip flops simultaneously.
- The change of state of a particular flip flop is dependent on the present state of other flip flops.
- For synchronous sequential circuits the design procedure is as follows:
 1. From the given information (word description/state diagram/timing diagram/other pertinent information) about the circuit, obtain the state table.
 2. Determine the number of flip flops needed.
 3. From the state table, derive the circuit excitation and output tables.
 4. Derive the circuit output functions and the flip flop input functions by simplification.
 5. Draw the logic diagram.

Ring Counter

- Circular shift register with only one flip flop being set at any particular time, all others are cleared.
- N bit ring counter will have N states and requires N flip flops.
- Fig. 11.2 shows a 4-bit ring counter using decoder and counter.




Figure 11.1 4-bit binary ripple counter




Figure 11.2 4-bit ring counter

Johnson counter or Switch tail ring counter

- Circular shift register with the complement output of the last flip flop connected to the input of the first flip flop.
- k-bit switch tail ring counter with 2k decoding gates provide outputs for 2k timing signals.

- k- bit Johnson counter requires k flip flops.
- Fig. 11.3 shows a 4-bit Johnson counter




Figure 11.3 4-bit Johnson counter

Solved Exercise

Write Verilog code for a 2-bit asynchronous up counter.

```

module tff1(T, Clock, Q);
    input T, Clock;
    output Q;
    reg Q;
    always @(negedge Clock)
    if (!T)
        Q <= Q;
    else
        Q <= ~Q;
endmodule

module twobit (clock, Q);
    input clock;
    output [1:0]Q;
    tff1 f1(1,clock,Q[0]);
    tff1 f2(1,Q[0],Q[1]);
endmodule

```

SIMPLE PROCESSOR DESIGN

Objectives:

In this lab student will be able to

1. Learn the concept of bus structure.

2. Understand the concept of a simple processor and a bit counting circuit.
3. Write Verilog code to implement bus structure, simple processor and bit counting circuit.

Bus Structure:

- When a digital system contains a number of n-bit registers, to transfer data from any register to any other register, a simple way of providing the desired interconnectivity is to connect each register to a common set of n wires, which are used to transfer data into and out of the registers. This common set of wires is usually called a bus.
- If common paths are used to transfer data from multiple sources to multiple destinations, it is necessary to ensure that only one register acts as a source at any given time and other registers do not interfere.

There are two arrangements for implementing the bus structure.

- Using Tri-State Drivers
- Using Multiplexers

Using Tri-State Drivers to Implement a Bus:

- Consider a system that contains k n-bit registers, R1 to Rk. Figure 12.1 shows how these registers can be connected using tri-state drivers to implement the bus structure. The data outputs of each register are connected to tri-state drivers. When selected by their enable signals, the drivers place the contents of the corresponding register onto the bus wires. The enable signals are generated by a control circuit.
- In addition to registers, in a real system, other types of circuit blocks would be connected to the bus. The figure shows how n bits of data from an external source can be placed on the bus, using the control input Extern.
- It is essential to ensure that only one circuit block attempts to place data onto the bus wires at any given time. The control circuit must ensure that only one of the tri-state driver enables signals, R1out, . . . , Rkout, is asserted at a given time. The control circuit also produces the signals R1in, . . . , Rkin, which determines when data is loaded into each register.
- In general, the control circuit could perform a number of functions, such as transferring the data stored in one register into another register and controlling the processing of data in various functional units of the system. Figure 12.1 shows an input signal named Function that instructs the control circuit to perform a particular task. The control circuit is synchronized by a clock input, which is the same clock signal that controls the k registers.




Fig 12.1 A digital system with k registers.

Solved Exercise

Consider a system that has three registers, $R1$, $R2$, and $R3$. The control circuit performs a single function—it swaps the contents of registers $R1$ and $R2$, using $R3$ for temporary storage. The required swapping is done in three steps, each needing one clock cycle. In the first step, the contents of $R2$ are transferred into $R3$. Then the contents of $R1$ are transferred into $R2$. Finally, the contents of $R3$, which are the original contents of $R2$, are transferred into $R1$. To transfer the contents of one register into another bus is used. The control circuit for this task can be explained in the form of a finite state machine as shown in Figure 12.2. Its state table is shown in Table 12.1




Fig. 12.2

Initially, the bus is loaded with the data when Extern=1. The initial state is A corresponding to load the data from the bus to register R1. To initiate the state transition from state A to state B, an input signal w is made equal to 1. In state A and state B, Extern = 1 to load two different data to the bus and then from the bus to the registers R1 and R2. The states B, C and D change their states if Extern = 0. From B, it goes to C state where R1 is copied to R3. From C it goes to state D where R2 is copied to R1. From D, the next state is E to copy R3 to R2 and an output Done = 1 to indicate that swap is completed. From E the next state is the initial state A. All transfers are taking place through the bus.

Table 12.1

PS	NS		Outputs								
	Extern=1	Extern=0	Rinext1	Rinext2	R1in	R1out	R2in	R2out	R3in	R3out	done
A	B	A	1	0	0	0	0	0	0	0	0
B	B	C	0	1	0	0	0	0	0	0	0
C	C	D	0	0	0	1	0	0	1	0	0
D	D	E	0	0	1	0	0	1	0	0	0
E	A	A	0	0	0	0	1	0	0	1	1

Verilog code

```
module regn (R, L, Clock, Q);  
parameter n = 8;  
input [n-1:0] R;  
input L, Clock;  
output [n-1:0] Q;  
reg [n-1:0] Q;  
always @(posedge Clock)  
if (L)  
Q <= R;  
endmodule // Code for 8-bit register
```

```
module swap1 (Resetn, Clock, w, Data, Extern, R1, R2, R3, BusWires, Done);  
parameter n = 8;  
input Resetn, Clock, w, Extern;  
input [n-1:0] Data;  
output [n-1:0] BusWires ,R1, R2, R3 ;  
reg [n-1:0] BusWires, R1, R2, R3;  
output Done;  
wire R1in, R1out, R2in, R2out, R3in, R3out, RinExt1, RinExt2;  
reg [2:0] y, Y;  
parameter [2:0] A = 3'b000, B = 3'b001, C = 3'b010, D = 3'b011, E = 3'b100;  
// Define the next state combinational circuit for FSM  
always @(w or y)  
begin  
case (y)  
A: if (w) Y = B;
```

```

else Y = A;

B: Y = C;

C: Y = D;

D: Y = E;

E: Y = A;

endcase

end

// Define the sequential block for FSM

always @(negedge Resetn or posedge Clock)

begin

if (Resetn == 0) y <= A;

else y <= Y;

end

// Define outputs of FSM

assign RinExt1 = (y == A);

assign RinExt2 = (y == B);

assign R3in = (y == C);

assign R1out = (y == C);

assign R2out = (y == D);

assign R1in = (y == D);

assign R3out = (y == E);

assign R2in = (y == E);

assign Done = (y == E);


always @(Extern or R1out or R2out or R3out)

if (Extern) BusWires = Data;

else if (R1out) BusWires = R1;

```

else if (R2out) BusWires = R2;

else if (R3out) BusWires = R3;

regn reg3 (BusWires, R3in, Clock, R3);

regn reg4 (BusWires, RinExt1 | R1in, Clock, R1);

regn reg5 (BusWires, RinExt2 | R2in, Clock, R2);

endmodule

Output:

